
Aufgabe

1. Mit der Maus wird ein Feld markiert.

2. Der Computer sucht nach den Schachregeln einen Weg, bei dem jedes Feld nur einmal besucht wird.

3. Es wird eine Antwort ausgegeben und der Lösungsweg aufgezeichnet.

procedure TryNext (i:integer; x,y:index; var q : boolean);

var

  k,u,v : integer;

  q1 : boolean;

begin

  k := 0;

  repeat

    k := k+1;

    q1 := false;

    u := x + a[k];

    v := y + b[k];

    if (u in s) and (v in s) then

      if h[u, v] = 0 then

        begin

          h[u,v] := i;

          if i < nsq then

            begin

              TryNext(i+1, u, v, q1);

              if not q1 then

                h[u, v] := 0;

            end

          else

            q1 := true;

        end;

  until q1 or (k=8);

  q := q1;

end;

[image: image2.png]Dame in Zeile
T % 1,5,8,6,3,7,2,4,
2 xX
3 xX
4 xX
5 xX
6 xX
7 xX
8 XK start
1.2 3 4 5 6 7 8 —
| SchlieBen





Aufgabe

1. Stelle 8 Damen so auf ein Schachbrett, dass keine Dame von einer anderen geschlagen werden kann.

2. Schreibe dazu das Programm.

Literatur: N. Wirth, Algorithmen S. 174

Methode

Nach den Regeln des Schachspiels kann in jeder Spalte und in jeder Zeile und in jeder Diagonalen nur 1 Dame stehen.

Man stellt die 1. Dame auf Spalte (i=1), Zeile (j=1). Damit sind diese Teile belegt. 

Man erhöht die Spalte auf (i=2) und die Zeile auf (j=2)  und setzt eine neue Dame. Wenn der Platz belegt ist (= false), wird die Zeile (j) erhöht.

procedure TForm1.VERSUCH (i: byte; var q: boolean);

var  j: byte;

begin

     for j:= 1 to 8 do begin

         if a[j] and b [i+j] and c[i-j] then begin                  // if ... = true wird abgefragt

              zeile[i]:= j;

              a[j]:= false;

              b[i+j]:= false;

              c[i-j]:= false;

              if i < 8 then begin

                  VERSUCH (i+1, q);              // Spalte erhöhen, nächste Dame setzen

                  if not q then begin

                      a[j]:= true;

                      b[i+j]:= true;

                      c[i-j]:= true;

                  end;

              end else q:= true;

         end;

     end

end;    // VERSUCH

procedure TForm1.Button1Click(Sender: TObject);

var i: byte;

    s: string;

begin

     stringGrid1.setFocus;   // dann verschwindet die blaue Markierung

     VERSUCH (1, q);

     for i:= 1 to 8 do begin

         s:= s + intTostr(zeile[i]) + ' , ';

         stringGrid1.cells[i, zeile[i]]:= 'XX';    // spalte, zeile

     end;

     listBox1.items.add(s);

end;

END.

Im Jahre 1848 stellte Max Bezzel aus Ansbach in der von der Berliner Schachgesellschaft herausgegebenen Zeitung die Aufgabe, acht Damen so auf ein Schachbrett zu stellen, daß keine von einer anderen geschlagen werden kann. Der Beitrag wurde so gut wie gar nicht beachtet. Erst als Franz Nauck aus Schleusingen das gleiche Problem zwei Jahre später in der Illustrierten Zeitung veröffentlichte, erregte es viel Interesse bei den Lesern. Selbst der Mathematiker Carl Friedrich Gauß beschäftigte sich mit der Aufgabe. Die Begeisterung für dieses Schachproblem war im letzten Jahrhundert so groß, daß es sogar in Form von Spielen herausgebracht wurde, zum Beispiel beim Schatz von Medinet oder beim Achterspiel.

Im Laufe der Zeit haben immer mehr Mathematiker den Reiz der Aufgabe entdeckt, die sie in erweiterter Form bis heute beschäftigt. Im Gegensatz zu den Schachspielern begnügten sie sich jedoch nicht mit einem gewöhnlichen Schachbrett mit 8 mal 8 Feldern. Sie wandten sich dem allgemeinen Fall zu, bei dem n Damen auf einem Brett mit n-n Feldern stehen, ohne daß eine von einer anderen bedroht wird, und fragten sich, wie viele verschiedene Anordnungen der Damen es zu jedem Wert von n gibt.

Es ist leicht einzusehen, daß nicht mehr als n Damen, die sich nicht gegenseitig bedrohen dürfen, auf ein n-n-Brett gestellt werden können. In jeder Spalte des Bretts darf nämlich höchstens eine Dame stehen. Stünde dort noch eine zweite, so könnte sie die erste schlagen. Da ein Schachbrett aber nur n Spalten hat, können auch nur n Damen auf das Brett gestellt werden. Will man genau n Damen auf das n-n-Brett setzen, so muss aus dem gleichen Grund in jeder Spalte und in jeder Reihe gerade eine Dame stehen.

Die Lösung für das 1-1-Schachbrett ist trivial. Eine einzelne Dame auf einem einfeldigen Brett kann natürlich von keiner anderen geschlagen werden. Beim 2-2- und beim 3-3-Brett sieht man sofort, daß es keine Lösungen gibt. Erst beim 4-4-Brett wird es interessanter. Es gibt zwei verschiedene Stellungen für die vier Damen, von denen jedoch die eine das Spiegelbild der anderen ist. Für das 5-5-Brett gibt es schon zehn verschiedene Lösungen, die sich aber alle aus zwei Grundlösungen herleiten lassen, indem man diese Anordnungen spiegelt oder um 90, 180 oder 270 Grad dreht. Überraschenderweise hat das größere 6-6-Brett nur vier Lösungen, die alle auf eine einzige Grundlösung zurückgehen. Mit wachsendem n steigt die Anzahl der Lösungen schnell an. Beim 7-7-Brett sind es 40 und beim normalen Schachbrett bereits 92 Lösungen. Gauß konnte übrigens das Acht-Damen-Problem nicht vollständig lösen: Er entdeckte nur 72 Stellungen. Bis 1890 hatte man schließlich das Damenproblem bis hin zu Brettgrößen von 11-11-Feldern in mühevoller und monatelanger Handarbeit gelöst. Beim 11-11-feldrigen Brett fand man 2680 verschiedene Lösungen.

Obwohl vom 7-7- bis zum 11-11-Schachbrett die Anzahl der möglichen Damenstellungen immer größer wurde, wußte niemand, ob es nicht doch noch Brettgrößen geben könnte, die keine Lösungen haben. Erst 1910 gelang es dem Mathematiker Wilhelm Ahrens, zu beweisen, daß es für jedes Schachbrett mit mehr als 3-3 Feldern Lösungen des Damenproblems gibt.

In den nächsten sechzig Jahren kam man mit dem Damenproblem nicht weiter voran. Beim 12-12- und beim 13-13-Schachbrett mußten so viele Stellungen der Figuren untersucht werden, daß dies von Hand selbst in jahrelanger Arbeit fast unmöglich war. A. Kopfermann aus Berlin versuchte es zwar, aber er verzählte sich. Sein Fehler wurden erst bemerkt, als der Mathematiker Mark B. Wells vom Los Alamos Scientific Laboratory in Neumexiko das Damenproblem mit dem Computer bearbeitete und 1971 die korrekten Zahlen veröffentlichte.

Mit dem Computer konnte man sich nun an immer größere Schachbretter wagen. Den Weltrekord halten zur Zeit die beiden französischen Informatiker Sylvain Pion und Joel Fourre. Sie errechneten 1998 mit Hilfe von 300 Computern, die gleichzeitig zwei Monate lang arbeiteten, daß es auf dem 23-23-Schachbrett 24 233 937 684 440 verschiedene Stellungen der 23 Damen gibt, die sich auf 3 029 242 658 210 Grundlösungen zurückführen lassen.

Das Damenproblem ist nicht nur eine Spielerei für Schachspieler und Mathematiker. Um es zu lösen, benutzt man in der Regel Backtracking-Algorithmen. Dies sind mathematische Verfahren, mit denen man kombinatorische Aufgaben systematisch durchprobiert und die sich dadurch auszeichnen, daß selbst sehr umfangreiche Probleme nur wenig Speicherplatz im Computer benötigen. Das Damenproblem hat sich nun in den letzten Jahrzehnten zu einem der Standardprobleme gemausert, mit dem in der Informatik Backtracking-Algorithmen und Computer getestet und miteinander verglichen werden. Die Folge davon ist eine wahre Flut von wissenschaftlichen Veröffentlichungen über die Schachdamen.

Auf einem n-n-Schachbrett kann eine Dame in horizontaler und in vertikaler Richtung jeweils n Felder bedrohen. Dabei ist das Feld mitgezählt, auf dem die Dame selbst steht. In den beiden diagonalen Richtungen sind es jedoch in der Regel weniger. Der Wert schwankt je nach Position der Dame zwischen einem Feld und n Feldern. Diese Asymmetrie stört das Ästhetikempfinden vieler Mathematiker. Deshalb stellen sie ihre Damen nicht nur auf ein gewöhnliches ebenes, sondern auch auf ein toroidales Schachbrett. 

Will man ein toroidales Schachbrett erhalten, kann man sich zunächst einmal ein ebenes Brett aus Gummi besorgen. Dieses wird nun zu einer Röhre zusammengerollt, so daß die Ober- und die Unterkante des Bretts aneinanderstoßen. Anschließend biegt man die Röhre zu einem Ring, wodurch die beiden Öffnungen der Röhre, die von der linken und rechten Kante des Schachbretts gebildet werden, aufeinandertreffen. Das Schachbrett sieht nun zwar aus wie ein Schwimmring, hat aber trotzdem immer noch n-n Felder. Auf diesem Schachbrett kann eine Dame, ganz egal wo sie steht, horizontal und vertikal sowie in beiden diagonalen Richtungen jeweils genau n Felder bedrohen. Eine Folge dieser größeren Bewegungsfreiheit der Damen ist, daß es für die verschiedenen Werte von n beim toroidalen Brett weniger Lösungen gibt als beim ebenen. Der ungarische Mathematiker George Pölya war der erste, der sich mit dem toroidalen Damenproblem befaßte. Er stellte 1918 fest, daß es nur dann Lösungen besitzt, wenn n weder durch 3 noch durch 4 teilbar ist. Die höhere Symmetrie des Torusbrettes verringert auch den Rechenaufwand, den man für die Lösungssuche treiben muß. Trotzdem ist er für größere Schachbretter immer noch unvorstellbar groß, und bis 1999 hatte man erst die Lösungen für Bretter mit jeweils 1-1, 5-5, 7-7, 11-11, 13-13, 17-17, 19-19 und 23-23 Feldern gefunden. Das nächst größere Brett, für das es Lösungen des toroidalen Damenproblems geben kann, hat 25-25 Felder. Unlängst gelang es endlich dem Mathematiker Matthias Engelhardt von der Firma Siemens in Nürnberg, dieses Problem in den Griff zu bekommen. Er fand 395 551 Grundlösungen, aus denen sich durch Drehungen, Spiegelungen und Verschiebungen des Torusbrettes 

957 725 000 Stellungen der 25 Damen ergeben.
	Felder

n * n
	Lösungen
	

	1
	1
	

	2
	-
	

	3
	-
	

	4
	2
	sie sind Spiegelbild von einander

	5
	10
	2 Grundlösungen, die anderen sind um 90, 180, 270 Grad gedreht

	6
	4
	erstaunlich wenige

	7
	40
	

	8
	92
	Gauss hatte 63 gefunden

	11
	2680
	

	23
	2,4 *1013
	Davon sind 3,0 *1012 Grundlösungen, die anderen erhält man durch Drehungen. Mit 30 Parallelrechnern hat man 1998 zwei Monate dafür gerechnet.

Aufgabe

Zeige die Lösungsstrategie auf, indem die nächste Dame gesetzt und nach Prüfung eventuell wieder entfernt wird.

procedure TForm1.VERSUCH (i: byte; var q: boolean);

var j: byte;

begin

     for j:= 1 to 8 do begin

         if a[j] and b [i+j] and c[i-j] then begin

              stringGrid1.cells[i, j]:= 'XX';              // Dame setzen

              PAUSE (500);

              zeile[i]:= j;

              a[j]:= false;

              b[i+j]:= false;

              c[i-j]:= false;

              if i < 8 then begin

                  VERSUCH (i+1, q);

                  if not q then begin

                      stringGrid1.cells[i, j]:= '  ';      // Dame löschen

                      PAUSE (500);

                      a[j]:= true;

                      b[i+j]:= true;

                      c[i-j]:= true;

                  end;

              end

              else q:= true;

         end;

     end;

end;    // VERSUCH

procedure TForm1.Button1Click(Sender: TObject);

var k: byte;

    s: string;

begin

    stringGrid1.setFocus;             // dann verschwindet die blaue Markierung

    VERSUCH (1, q);

    for k:= 1 to 8 do s:= s+ intTostr(zeile[k]) + ' , ';

    listBox1.items.add(s);

end;

END.

Aufgabe

Es sind alle Lösungen des Damenproblems aus zu geben.

procedure TForm1.initialisierung;

var k: integer;

begin

    for k:= 1 to 8 do stringGrid1.cells[0, k] := ' ' + intTostr(k); //  Zeilen-Nummer

//    for k:= 1 to 8 do stringGrid1.cells[k, 9] := ' ' + intTostr(k); // Nummer Spalten

    for k:= 1 to 8  do a[k]:= true;     // alle Zeilen als frei markieren

    for k:= 2 to 16 do b[k]:= true;     // die Diagonale (/) als frei markieren

    for k:= -7 to 7 do c[k]:= true;     // die Diagonale (\) als frei markieren

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

     INITIALISIERUNG;

end;

procedure TForm1.VERSUCH (i: byte);

var j, x: byte;

begin

     for j:= 1 to 8 do begin

        if a[j] and b[i+j] and c[i-j] then begin

              zeile[i]:= j;

              a[j]:= false;

              b[i+j]:= false;

              c[i-j]:= false;

              if i < 8 then VERSUCH (i+1) else AUSGABE;

              a[j]:= true;

              b[i+j]:= true;

              c[i-j]:= true;

        end;

     end;

end;    // VERSUCH

procedure TForm1.AUSGABE;

var s: string;

    k: byte;

begin

    for k:= 1 to 8 do s:= s+ intTostr(zeile[k]) + ' , ';

        listBox1.items.add(s);

end;

procedure TForm1.Button1Click(Sender: TObject);

begin

    stringGrid1.setFocus;             // dann verschwindet die blaue Markierung

    VERSUCH (1);

end;

END.

PROGRAMM  FISCHE

HMO ludens, Warendorf, 27.6.1988, ein Beispiel für  BACKTRACKING.

M.C.Escher hat auf einer Kugeloberfläche 12 Fische so angeordnet, daß sie

eine Parkettierung der Oberfläche bilden. Die Kugel von Escher haben Doris

Schattschneider und Wallace Walker zu einem Würfel deformiert, der mit dem

gleichen Fischmuster bedeckt ist. Viermal stoßen 3 Fische jeweils mit den

Schwanzflossen zusammen und viermal stoßen 3 Fische jeweils mit der Rücken-

flosse zusammen. Dieses raffinierte Muster muß man sehen !

Das Muster soll jetzt so gefärbt werden, daß

-  jeder Fisch eine Farbe erhält,

-  benachbarte Fische unterschiedliche Farben haben,

-  genau vier Farben verwendet werden.

Bei 12 Fischen in 4 Farben liefert die Kombinatorik   4^12 = 16.777.216

Möglichkeiten. Darunter dürften die meisten nicht die Bedingungen erfüllen.

Es ergeben sich genau 240 Lösungen, von den aber die meisten zueinander

symmetrisch sind. }

CONST N               =  12;    { Gegeben sind 12 Fische }

      farbenanzahl    =  4;     { Gesucht ist eine Lösung mit 4 Farben }

      auswahlanfang   =  1;

      auswahlende     =  farbenanzahl;

TYPE  inzidenztyp     =  SET OF 0..N;

      vektortyp       =  ARRAY [0..N] OF 0..farbenanzahl;

CONST {Aus dem vorgelegten Muster der 12 Fische, ergibt sich, daß jeder Fisch

       5 andere berührt. Die Fische werden (willkürlich) mit Nummern von 1 bis

       12 markiert. In der Menge "inzidenzen[i]" sind alle die Nummern der

       Fische enthalten, die den Fisch mit Nummer i berühren.}

      inzidenzen      :  ARRAY [1..12] OF inzidenztyp

                      = ([2,3,4,5,6],   [1,3,6,7,8],   [1,2,5,8,9],

                         [1,5,6,10,11], [1,3,4,9,10],  [1,2,4,7,11],

                         [2,6,8,11,12], [2,3,7,9,12],  [3,5,8,10,12],

                         [4,5,9,11,12], [4,6,7,10,12], [7,8,9,10,11]);

      farbbezeichnung :  ARRAY [1..farbenanzahl] OF String[4]

                      = ('rot', 'blau', 'gelb', 'grün');

VAR   zaehler         :  Integer;

      vektor          :  vektortyp;

  { T }

  PROCEDURE try (stufe: Integer; vektor: vektortyp);

  VAR wahl         : Integer;

      lebend       : Boolean;

      neues_optimum: Boolean;

    { T.1 }

    PROCEDURE vektor_bewerten;

    { Es muß nur geprüft werden, ob die Farbe bei Position k im Vektor

      nicht gleich den Farben der benachbarten bereits gesetzten Fische ist. }

    VAR i,k: Integer;

    BEGIN

      k:=vektor[0]; lebend:=TRUE; i:=1;

      WHILE (i<k) AND lebend DO

      BEGIN

        IF (vektor[i] = vektor[k]) AND (i IN inzidenzen[k])

           THEN lebend:=FALSE;

        i:=i+1

      END;

      neues_optimum:=lebend

    END;

    { T.2 }

    FUNCTION vollstaendig: Boolean;

    BEGIN  vollstaendig := (vektor[0]=N)  END;

    { T.3 }

    PROCEDURE loesung_speichern;

    { Speichern entfällt, stattdessen wird eine Lösung ausgegeben }

    VAR i: Integer;

    BEGIN

      zaehler:=zaehler+1; { zaehler für die Lösungen, global! }

      Write(zaehler:2,' : ');

      FOR i:=1 TO N DO Write(farbbezeichnung[vektor[i]]:5); Writeln

    END;

  BEGIN { T }

    FOR wahl:=auswahlanfang TO auswahlende DO

    BEGIN

      vektor[0]:=stufe;      { Stelligkeit setzen, Fisch "stufe" ist dran }

      vektor[stufe]:=wahl;   { wahl für diesen Fisch setzen }

      vektor_bewerten;       { hier feststellen ob Farbkonflikt }

      IF vollstaendig

        THEN BEGIN  IF neues_optimum THEN  loesung_speichern   END

        ELSE BEGIN  IF lebend        THEN  try(stufe+1,vektor) END

    END

  END;

BEGIN

  zaehler:=0;

  Writeln; Writeln('P I S C E S   E S C H E R I');

           Writeln('---------------------------'); Writeln;

  vektor[0]:=0;

  try(1,vektor)

END.

PROGRAMM MÜNZEN

HMO ludens, Warendorf, 27.6.1988, ein Beispiel für  BACKTRACKING.

Teilmengenproblem: Gegeben sind N Münzen mit den Münzwerten

p1 , p2 , .. , pN  und ein Betrag B. Gesucht sind alle Möglichkeiten,

den Betrag B mit möglichst wenig Münzen auszuzahlen. Bei dieser

Übungsaufgabe k”önen die Daten eingegeben werden. }

CONST  Nmax          =  50;         { vorgesehen f�r maximal 50 M�nzen }

       auswahlanfang =  0;

       auswahlende   =  1;

TYPE   vektortyp     =  ARRAY [0..Nmax] OF Integer;

       inhaltstyp    =  RECORD      { Typ f�r einen Listeneintrag }

                          vektor: vektortyp;

                          anzahl: Integer

                        END;

{$I stackadt.pas}

VAR    pvektor       :  ARRAY [1..Nmax] OF Integer; { Werte }

       vektor        :  vektortyp;

       N             :  Integer;    { Anzahl der M�nzen }

       B             :  Integer;    { Zielbetrag }

       i             :  Integer;

       akt_optimum   :  Integer;

       loesungen     :  stacktyp;

       eintrag       :  inhaltstyp;

  { T }

  PROCEDURE try (stufe: Integer; vektor: vektortyp);

  VAR wahl,anzahl,wert: Integer;

    { T.1 }

    PROCEDURE vektor_bewerten;

    VAR j: Integer;

    BEGIN

      anzahl:=0; wert:=0;

      FOR j:=1 TO vektor[0]

      DO BEGIN anzahl:=anzahl+vektor[j];

               wert:=wert+vektor[j]*pvektor[j]

         END

    END;

    { T.2 }

    FUNCTION vollstaendig: Boolean;

    BEGIN  vollstaendig := (vektor[0]=N)  END;

    { T.3 }

    FUNCTION neues_optimum: Boolean;

    BEGIN  neues_optimum := (wert=B) AND (anzahl<=akt_optimum)  END;

    { T.4 }

    FUNCTION lebend: BOOLEAN;

    BEGIN  lebend := (anzahl<=akt_optimum) AND (wert<=B)  END;

    { T.5 }

    PROCEDURE loesung_speichern;

    VAR eintrag: inhaltstyp;

    BEGIN

      akt_optimum:=anzahl;

      eintrag.vektor:=vektor; eintrag.anzahl:=anzahl;

      push_stack(loesungen,eintrag)

    END;

  BEGIN { T }

    FOR wahl:=auswahlanfang TO auswahlende DO

    BEGIN

      vektor[0]:=stufe;        { Stelligkeit setzen }

      vektor[stufe]:=wahl;     { Auswahl Objekt i trffen }

      vektor_bewerten;         { Anzahl der M�nzen, M�nzwert berechnen }

      IF vollstaendig

      THEN BEGIN IF neues_optimum THEN  loesung_speichern   END

      ELSE BEGIN IF lebend        THEN  try(stufe+1,vektor) END

    END

  END;

  PROCEDURE ein (was: String; unten,oben: Integer; VAR wert: Integer);

  VAR s: String[10];

      c: Integer;

  BEGIN

    REPEAT

      write(was);

      readln(s); IF s='' THEN s:='x';

      val(s,wert,c);

    UNTIL (c=0) AND (wert>=unten) AND (wert<=oben)

  END;

BEGIN

  { šberschrift, kurze Hinweise, Eingabe der Daten }

  Writeln; Writeln('BACKTRACKING :  Teilmengenproblem');

           Writeln('---------------------------------');

  Writeln('Gegeben sind N=',n,' M�nzen mit beliebigen Werten.');

  Writeln('und ein Zielbetrag B. Gesucht sind alle M”glichkeiten den');

  Writeln('Betrag B mit m”glichst wenig M�nzen auszuzahlen.'); Writeln;

  ein('Anzahl M�nzen  N = ',1,50,N);

  ein('Zielwert       B = ',0,Maxint,B);

  FOR i:=1 TO N DO

  BEGIN write(i:2,'-'); ein('Wert        P = ',0,Maxint,pvektor[i]) END;

  { Initialisierung, L”sungssuche }

  create_stack(loesungen); { Liste f�r L”sungen initialisieren }

  vektor[0]:=0;            { Start mit 0-stelligem Vektor }

  akt_optimum:=N+1;        { Ergebnis N+1 1 M�nzen bedeutet keine L”sung }

  try(1,vektor);           { rekursive L”sungssuche }

  { Ausgabe der L”sungen }

  writeln;

  IF akt_optimum>N

    THEN writeln('keine L”sung !')

    ELSE BEGIN

      WHILE NOT empty_stack(loesungen)

      DO BEGIN

        top_stack(loesungen,eintrag); pop_stack(loesungen);

        IF eintrag.anzahl=akt_optimum

        THEN BEGIN

          FOR i:=1 TO N DO

            IF eintrag.vektor[i]=0 THEN Write('-':6)

                                   ELSE Write(pvektor[i]:6);

          Writeln;

        END

      END

    END

END.

PROGRAMM RUCKSACK

HMO ludens, Warendorf, 27.6.1988, ein Beispiel f�r  BACKTRACKING.

Rucksackproblem: Gegeben sind N Objekte mit den Werten  p1 , .. , pN

und den Gewichten  w1, .. , wN  und ein Wert B f�r das maximale

Gewicht.  Gesucht ist eine Auswahl von Objekten, bei denen das Gewicht

B nicht �berschritten wird und der Gesamtwert maximal ist.

Bei dieser šbungsaufgabe sind die Daten als Konstanten vorgegeben. }

CONST  N             =  10;    { Gegeben sind 5 M�nzen }

       B             =  200;   { Zielbetrag }

       auswahlanfang =  0;     { Objekt nicht nehmen }

       auswahlende   =  1;     { Objekt nehmen }

TYPE   vektortyp     =  ARRAY [0..N] OF Integer;

CONST  pvektor       :  ARRAY [1..N] OF Integer =      { Werte }

                        (100, 70, 60, 40, 35, 30, 30, 25, 20, 5);

       wvektor       :  ARRAY [1..N] OF Integer =      { Gewichte }

                        (110, 95, 85, 80, 55, 50, 45, 35, 20, 10);

VAR    vektor        :  vektortyp;

       i             :  Integer;

       akt_optimum   :  Integer;

       loesung       :  vektortyp;

  { T }

  PROCEDURE try (stufe: Integer; vektor: vektortyp);

  VAR wahl,gewicht,wert: Integer;

    { T.1 }

    PROCEDURE vektor_bewerten;

    VAR j: Integer;

    BEGIN

      wert:=0; gewicht:=0;

      FOR j:=1 TO vektor[0]

        DO BEGIN  gewicht:= gewicht+ vektor[j]*wvektor[j];

                  wert   := wert   + vektor[j]*pvektor[j]

           END

    END;

    { T.2 }

    FUNCTION vollstaendig: Boolean;

    BEGIN  vollstaendig := (vektor[0]=N)  END;

    { T.3 }

    FUNCTION neues_optimum: Boolean;

    BEGIN  neues_optimum := (gewicht<=B) AND (wert>akt_optimum)  END;

    { T.4 }

    FUNCTION lebend: BOOLEAN;

    BEGIN  lebend := (gewicht<=B)  END;

    { T.5 }

    PROCEDURE loesung_speichern;

    BEGIN  akt_optimum:=wert;  loesung:=vektor  END;

  BEGIN { T }

    FOR wahl:=auswahlanfang TO auswahlende DO

    BEGIN

      vektor[0]:=stufe;     { Stelligkeit setzen }

      vektor[stufe]:=wahl;  { Auswahl Objekt i treffen }

      vektor_bewerten;

      IF vollstaendig

        THEN BEGIN IF neues_optimum THEN  loesung_speichern   END

        ELSE BEGIN IF lebend        THEN  try(stufe+1,vektor) END

    END

  END;

BEGIN

  { šberschrift und kurze Hinweise }

  Writeln; Writeln('BACKTRACKING :  knapsac');

           Writeln('-----------------------');

  Writeln('Gegeben sind N=',N,' Objekte mit den Werten:');

  FOR i:=1 TO n-1 DO Write(pvektor[i],', ');

  Writeln(pvektor[i]);

  Writeln('und den Gewichten:');

  FOR i:=1 TO n-1 DO Write(wvektor[i],', ');

  Writeln(wvektor[i]);

  Writeln('und ein Betrag B=',B,'. Gesucht ist eine Auswahl der Objekte,');

  Writeln('deren Gewicht B nicht �berschreitet und deren Wert maximal ist.');

  Writeln;

  { Initialisierung, L”sungssuche }

  vektor[0]:=0;      { Start mit 0-stelligem Vektor }

  akt_optimum:=0;    { Ergebnis 0 bedeutet keine L”sung }

  try(1,vektor);     { rekursive L”sungssuche }

  { Ausgabe der L”sung }

  IF akt_optimum=0

    THEN writeln('keine L”sung !')

    ELSE BEGIN

      Write('Werte:    ');

      FOR i:=1 TO N DO

        IF loesung[i]=0 THEN Write('-':6)

                        ELSE Write(pvektor[i]:6);

      Writeln;

      Write('Gewichte: ');

      FOR i:=1 TO N DO

        IF loesung[i]=0 THEN Write('-':6)

                        ELSE Write(wvektor[i]:6);

      Writeln;

      Writeln('maximaler Wert: ',akt_optimum); Writeln

    END

END.

PROGRAMM SALESMAN

Hubert Deitemann, WAF, Dez.1997

Gegeben sind N Orte, die miteinander verbunden sind. In dieser Variante

sind die Orte symmetrisch miteinander verbunden, d.h. es gibt keine

Einbahnstraáen. Es d�rfen aber Verbindungen fehlen. Gesucht ist eine

k�rzeste Rundreise, die jeden Ort genau einmal besucht und zum Ausgangsort

zur�ckkehrt.}

CONST Nmax          = 20;                           {max. Zahl von Orten}

TYPE  vektortyp     = ARRAY [-1..Nmax] OF Integer;  {eine Rundreise

                         -1: Bewertung des Vektors, akkum.Streckenl„ngen

                          0: Stelligkeit des Vektors, fertig wenn gleich N

                       1..N: Nummern der Orte in Reihenfolge des Besuchens}

      inzidenztyp   = ARRAY [1..Nmax,1..NMax] OF Integer;{Entfernungen}

      wahltyp       = SET OF 1..Nmax;               {Menge f�r besuchte Orte}

CONST auswahlanfang : Integer = 1;

      standard      : inzidenztyp =

                     ((  0, -1, -1,221, 51, 76, 35, 49, 80, 97, 83,0,0,0,0,0,0,0,0,0),

                      ( -1,  0, -1,218, 21,150, 85, -1, 52,135, 38,0,0,0,0,0,0,0,0,0),

                      ( -1, -1,  0, -1, -1, 98,  5, 59,205,162, -1,0,0,0,0,0,0,0,0,0),

                      (221,218, -1,  0,253, -1,100, 42,107,177, 16,0,0,0,0,0,0,0,0,0),

                      ( 51, 21, -1,253,  0, 96,180, 65,172, 56, -1,0,0,0,0,0,0,0,0,0),

                      ( 76,150, 98, -1, 96,  0,245, 23, -1, 18, -1,0,0,0,0,0,0,0,0,0),

                      ( 35, 85,  5,100,180,245,  0, 17, -1,156, -1,0,0,0,0,0,0,0,0,0),

                      ( 49, -1, 59, 42, 65, 23, 17,  0,179,202, 35,0,0,0,0,0,0,0,0,0),

                      ( 80, 52,205,107,172, -1, -1,179,  0,203,165,0,0,0,0,0,0,0,0,0),

                      ( 79,135,162,177, 56, 18,156,202,203,  0,174,0,0,0,0,0,0,0,0,0),

                      ( 83, 38, -1, 16, -1, -1, -1, 35,165,174,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0),

                      (  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,0,0,0,0,0,0,0,0,0));

VAR   N             : Integer;       {konkrete Anzahl der Orte, N<=Nmax}

      auswahlende   : Integer;       {ist gleich N}

      inzidenz      : inzidenztyp;   {Entfernungen}

      vektor        : vektortyp;

      gewaehlt      : wahltyp;

      akt_optimum   : Integer;

      loesung       : vektortyp;

      i, j          : Integer;

      c             : Char;

  PROCEDURE zahlein (VAR zahl: Integer; u,o: Integer; hinw: STRING);

  VAR s: STRING[5];                     {Zahleneingabeprozedur mit}

      c: Integer;                       {Plausibilit„tskontrollen}

  BEGIN

    REPEAT

      Write(hinw);  Readln(s);          {Hinweistext, String einlesen}

      IF s='' THEN c:=1 ELSE Val(s,zahl,c); {String s in Zahl umwandeln}

      IF (c<>0) OR (zahl<u) OR (zahl>o) {keine Zahl, auáerhalb Bereich}

         THEN Writeln('Eingabefehler !')

    UNTIL (c=0) AND (zahl>=u) AND (zahl<=o) {bis die Zahl OK ist}

  END;

  { T }

  PROCEDURE try (stufe: Integer; vektor: vektortyp; gewaehlt: wahltyp);

  VAR wahl     : Integer;

      lebend   : Boolean;

      neuwahl  : wahltyp;

      neuvektor: vektortyp;

    { T.1 }

    FUNCTION vollstaendig: Boolean;

    BEGIN  vollstaendig := (neuvektor[0]=N)  END;

    { T.2 }

    PROCEDURE vektor_bewerten;

    VAR strecke: Integer;

    BEGIN

      IF wahl IN gewaehlt                {tot, weil gew„hlter Ort bereits besucht}

        THEN lebend:=FALSE

        ELSE BEGIN

          lebend:=True;                  {gew„hlter Ort erstmal noch m”glich}

          neuwahl:=gewaehlt;

          neuwahl:=neuwahl+[wahl];       {in Auswahlmenge Ort hinzuf�gen}

          IF stufe=1

            THEN neuvektor[-1]:=0

            ELSE BEGIN

              strecke:=inzidenz[neuvektor[stufe-1],neuvektor[stufe]];

              IF strecke<=0

                THEN lebend:=False       {tot, weil keine Verbindung}

                ELSE BEGIN

                  neuvektor[-1]:=neuvektor[-1]+strecke;

                  {Bewertung um neue Strecke erg„nzen}

                  lebend:=(neuvektor[-1]<akt_optimum)

                  {auch tot, wenn Strecke schon l„nger als bisheriges Optimum}

                END

            END

        END

    END;

    { T.3 }

    FUNCTION neues_optimum: Boolean;

    VAR strecke: Integer;

    BEGIN

      IF NOT lebend

        THEN neues_optimum:=False  {wenn tot, dann erst recht kein Optimum}

        ELSE BEGIN

          strecke:=inzidenz[neuvektor[n],neuvektor[1]];

          IF strecke<=0

            THEN neues_optimum:=False {zur�ck zum Start geht nicht}

            ELSE BEGIN

              IF ((neuvektor[-1]+strecke)>=akt_optimum)

                THEN neues_optimum:=False {Strecke zu lang}

                ELSE BEGIN

                  neuvektor[-1]:=neuvektor[-1]+strecke;

                  neues_optimum:=True

                END

            END

        END

    END;

    { T.4 }

    PROCEDURE loesung_speichern;

    BEGIN

      akt_optimum:=neuvektor[-1];

      loesung:=neuvektor

    END;

  BEGIN { T }

    FOR wahl:=auswahlanfang TO auswahlende DO

    BEGIN

      neuvektor:=vektor;

      neuvektor[0]:=stufe;           {Stelligkeit setzen}

      neuvektor[stufe]:=wahl;        {Auswahl Objekt i treffen}

      vektor_bewerten;

      IF vollstaendig

        THEN BEGIN IF neues_optimum THEN loesung_speichern              END

        ELSE BEGIN IF lebend        THEN try(stufe+1,neuvektor,neuwahl) END

    END

  END;

BEGIN

  { šberschrift und kurze Hinweise, Eingaben }

  Writeln; Writeln ('BACKTRACKING: Travelling Salesman');

           Writeln ('---------------------------------');

  Write('E(ingabe oder S(tandardbeispiel ');

  REPEAT Readln(c) UNTIL c IN ['e','E','s','S',#13]; Writeln;

  IF c IN ['S','s']

  THEN BEGIN N:=11; inzidenz:=standard END

  ELSE BEGIN

    zahlein (N, 2, Nmax, 'Anzahl der Orte: ');

    Writeln('Entfernungen als ganze Zahlen eingeben. Falls keine');

    Writeln('Verbindung existiert, bitte -1 eingeben.');

    FOR i:=1 TO N-1 DO

    BEGIN

      inzidenz[i,i]:=0;

      FOR j:=i+1 TO N DO

      BEGIN

        write(i,'-',j,'-');

        zahlein (Inzidenz[i,j],-1,999,'Entfernung : ');

        Inzidenz[j,i]:=Inzidenz[i,j]

      END

    END;

    Writeln

  END;

  FOR i:=1 TO N DO

  BEGIN

    write(i:2,': ');

    FOR j:=1 TO N DO

      IF inzidenz[i,j]<0 THEN write('   -') ELSE write(Inzidenz[i,j]:4);

    Writeln

  END;

  { Initialisierung, L”sungssuche }

  auswahlende:=N;

  akt_optimum:=Maxint;

  gewaehlt:=[];

  vektor[0]:=0;

  vektor[-1]:=0;

  try (1,vektor,gewaehlt);

  { Ausgabe }

  Writeln;

  IF akt_optimum=Maxint THEN Writeln('Geht nicht !')

  ELSE  BEGIN

    Writeln('Entfernung: ',akt_optimum);

    FOR i:=1 TO N DO Write(loesung[i],', '); Writeln(loesung[1])

  END

END.

PROGRAMM WIMPEL

bunte_reihe;  { 15. Bundeswettbewerb Informatik 1996/97

1.Aufgabe, Hubert Deitemann, Warendorf, 23.9.1996

Gegeben sind N Wimpel, die in M verschiedenen Farben gef„rbt sind. Jede

Farbe kommt mit einer bestimmten Anzahl vor. Gesucht sind Anordnungen

der Wimpel zu einer Reihe, bei der benachbarte Wimpel unterschiedliche

Farben haben.

Beispiel: N=4, M=2;  jede Farbe kommt 2x vor

          Es gibt zwei L”sungen:  1.   1 2 1 2

                                  2.   2 1 2 1

          Die Reihe  1 1 2 2  ist z.B. keine L”sung

Es soll die Anzahl aller L”sungen bei einer bestimmten Eingabe ausgegeben

werden. Wenn bis zu 10 L”sungen existieren, sollen alle L”sungen ausgegeben

werden; existieren mehr L”sungen, dann nur die ersten 3.

Eingabedaten f�r ein Beispiel:  N=10, M=5, jede Farbe kommt 2x vor. }

CONST Nmax          = 20;                           {max. Zahl von Wimpeln}

      Mmax          = 10;                           {max. Zahl der Farben}

      Wmax          = 10;                           {max. Zahl Wimpel in einer Farbe }

TYPE  vektortyp     = ARRAY [0..Nmax] OF Integer;   {eine "bunte Reihe"}

      vorratstyp    = ARRAY [1..Mmax] OF Integer;   {Wimpel je Farbe}

      farbentyp     = ARRAY [1..Mmax] OF STRING[10];{Name jeder Farbe}

      loesungtyp    = ARRAY [1..10]   OF vektortyp; {bis zu 10 "Reihen"}

CONST auswahlanfang : Integer = 1;

VAR   auswahlende   : Integer;                      {ist gleich M}

      N             : Integer;                      {Anzahl der Wimpel}

      M             : Integer;                      {Anzahl der Farben}

      zaehler       : longint;                      {Anzahl der L”sungen}

      vorrat        : vorratstyp;                   {Anzahl der Wimpel je Farbe}

      farben        : farbentyp;                    {Name jeder Farbe}

      loesung       : loesungtyp;                   {bis zu 10 "Reihen"}

      vektor        : vektortyp;

      i, j          : Integer;

  PROCEDURE zahlein (VAR zahl: Integer; u,o: Integer; hinw: STRING);

  VAR s: STRING[5];                     {Zahleneingabeprozedur mit}

      c: Integer;                       {  Plausibilit„tskontrollen}

  BEGIN

    REPEAT

      Write(hinw);  Readln(s);          {Hinweistext, String einlesen}

      IF s='' THEN c:=1 ELSE Val(s,zahl,c); {String s in Zahl umwandeln}

      IF (c<>0) OR (zahl<u) OR (zahl>o) {keine Zahl, auáerhalb Bereich}

         THEN Writeln('Eingabefehler !')

    UNTIL (c=0) AND (zahl>=u) AND (zahl<=o) {bis die Zahl OK ist}

  END;

  { T }

  PROCEDURE try (stufe: Integer; vektor: vektortyp; vorrat: vorratstyp);

  VAR wahl         : Integer;

      neues_optimum: Boolean;

      lebend       : Boolean;

      neuvorrat    : vorratstyp;

    { T.1 }

    PROCEDURE vektor_bewerten;

    BEGIN

      IF vorrat[wahl]<=0       { gew„hlte Farbe nicht mehr im Vorrat }

        THEN lebend:=FALSE

        ELSE BEGIN

          neuvorrat:=vorrat;

          neuvorrat[wahl]:=neuvorrat[wahl]-1; { vom Farbvorrat entfernen }

          lebend:=(stufe=1) OR (vektor[stufe]<>vektor[stufe-1])

        END;

        neues_optimum:=lebend

    END;

    { T.2 }

    FUNCTION vollstaendig: Boolean;

    BEGIN  vollstaendig := (vektor[0]=N)  END;

    { T.3 }

    PROCEDURE loesung_speichern;

    BEGIN

      zaehler:=zaehler+1;

      IF zaehler<11 THEN loesung[zaehler]:=vektor

    END;

  BEGIN { T }

    FOR wahl:=auswahlanfang TO auswahlende DO

    BEGIN

      vektor[0]:=stufe;           { Stelligkeit setzen }

      vektor[stufe]:=wahl;        { Auswahl Objekt i treffen }

      vektor_bewerten;

      IF vollstaendig

        THEN BEGIN IF neues_optimum THEN  loesung_speichern             END

        ELSE BEGIN IF lebend        THEN  try(stufe+1,vektor,neuvorrat) END

    END

  END;

BEGIN

  { šberschrift und kurze Hinweise, Eingaben }

  Writeln; Writeln ('15.BW Inf., Aufgabe 1, "Bunte Reihe"'); Writeln;

  zahlein (M, 1, Mmax, 'Anzahl der Farben: ');

  auswahlende:=M;

  FOR i := 1 TO M DO

  BEGIN

    Write('Farbe ',i,': ');

    Readln(farben[i])                   {jeden Farbnamen eingeben}

  END; {FOR i}

  N:=0;                                 {wird die Gesamtzahl der Wimpel}

  FOR i := 1 TO M DO

  BEGIN                                 {pro Farbe die Anzahl eingeben}

    zahlein(vorrat[i],1,Wmax,'Anzahl der '+farben[i]+'en Wimpel: ');

    N:=N+vorrat[i];                     {Gesamtzahl inkrementieren}

  END; {FOR i}

  IF N>Nmax THEN                        {zuviele Wimpel ?}

  BEGIN

    Writeln('Obergrenze �berschritten, Abbruch!');

    Halt                                {Notausgang}

  END;

  { Initialisierung, L”sungssuche }

  zaehler:=0;

  vektor[0]:=0;

  try (1,vektor,vorrat);

  { Ausgabe }

  Writeln;

  IF zaehler=0 THEN Writeln('Geht nicht !') {keine L”sung}

  ELSE  BEGIN                            {mindestens 1 L”sung}

    IF zaehler>10 THEN

    BEGIN                               {mehr als 10 L”sungen}

      Writeln(zaehler,' L”sungen');     {Anzahl der L”sungen ausgeben}

      Write('Beispiele f�r ');

      zaehler:=3                        {Z„hler auf 3 kappen}

    END; {IF loesanzz>10}

    Writeln('Reihenfolgen:');

    FOR j:=1 TO zaehler DO

    BEGIN

      Write(j:2,'. ');                  {Nummer der L”sung ausgeben}

      FOR i:=1 TO N DO Write(farben[loesung[j][i]],' ');

      Writeln

    END;

  END

END.

Aufgabe

1. Zeichne ein Labyrinth.

2. Mit der Maus wird ein Anfangsfeld markiert.

3. Der Computer sucht ab dieser Stelle den Weg aus dem Labyrinth.

procedure FINDEN (s, z: integer);         // s = spalte; z = zeile

begin

     MARKIEREN (s, z, marke);

     if rand(s-1, z) or rand(s, z-1)= true

       then

         begin

           form1.Label1.caption:=('  Ausgang gefunden  ');

           pause(1000);

           form1.Label1.caption:=('');

         end

     else begin

         if form1.stringGrid1.cells[s-1,z] = l then finden(s-1,z);

         if form1.stringGrid1.cells[s,z-1] = l then finden(s,z-1);

         if form1.stringGrid1.cells[s+1,z] = l then finden(s+1,z);

         if form1.stringGrid1.cells[s,z+1] = l then finden(s,z+1);

     end;

     MARKIEREN(s, z, l);

end;

